
나를 돋보이게하는 엑셀 자동화 서식만들기

(인사/총무편)

최 준 선

오피스튜터강사

1부. 연봉일람표자동화하기

강좌소개

서식업무는직장인이면누구나하는작업이지만자동화되지않아

도 불편하다는 생각은 별로 하지않는다.

실질적인 불편함에도불구하고습관적으로수동처리되어온엑셀

업무가있다면 반드시 자동화해볼 필요가 있다.

본 강좌를통해 업무자동화의 경험을쌓음으로써, 향후 보다 큰업

무도 자동화시킬 수있다는 자신감을얻을 수있다.

1부. 연봉일람표자동화하기

목차 (1)

1부. 연봉일람표 자동화하기

고급필터를이용해연봉일람표를 자동으로 구축할 수있는 서식예제안내

2부. 연봉계약서 자동화하기

워드의 ‘편지병합’ 기능을 이용한 서식 출력 작업 예제 안내

3부. 인건비분석표자동화하기

연봉일람표서식을소스로 기업의 인건비를 분석할 수 있는 서식작업안내

4부. 팀인사고과표자동화하기

조직내구성원을 평가하기 위해사용하는 서식을자동화하는 방법론에 대해 안내

5부. 경력증명서 자동화하기

직원명부와부서이동기록표를바탕으로 퇴직사원의경력증명서를 자동으로

구성하는 서식작성예제 안내

1부. 연봉일람표자동화하기

목차 (2)

6부. 출장명령서 자동화하기

출장인원을결정하면선지급될여비가등급에따라자동계산되는서식작성예제안내

7부. 근로계약서 자동화하기

계약직원명부와연동되는근로계약서서식작성예제안내

8부. 계약직원출/퇴근 관리 및임금계산표 자동화하기

계약직원의출/퇴근시간기록을자동화하고, 기록된출/퇴근시간을통해지불할임금내역을

피벗테이블로집계하는서식작성예제안내

9부. 연수원 방자동 배정표 자동화하기

연수원참가자들의방을자동으로배정하는서식작성예제안내

10부. 주간업무계획 자동화하기

주간단위로일정을정리하고, 정리된일정을아웃룩에자동등록할수있는서식 작성

예제안내

1부. 연봉일람표자동화하기

최 준 선

오피스튜터강사

1부. 연봉일람표자동화하기

1. 도입말 (1)

강좌목적

연봉일람표의구성을필요에 따라전체직원 또는 특정 조건과 일

치하는직원들의 내역으로 자유자재로 구성할 수있다.

연봉 협상 전직위별, 또는개인별인상률이 확정되면, 새로계약

할 연봉내역을 연봉일람표에 자동으로 표시되게된다.

강좌소개

고급필터를이용해원하는 조건의 직원 연봉내역만 추출해 낸다.

수식, 조건부서식, 매크로기록기등을이용해, 매번서식을입력

하고 수정하는 과정을 자동화한다. 

1부. 연봉일람표자동화하기

1. 도입말 (2)

강좌내용

연봉일람표기능 소개

직원명부이해하기

연봉표이해하기

연봉일람표작업 순서

근속년수계산하기

인상률에기반한 연봉 계산하기

퇴직 현황 표시하기

고급필터를이용해원하는 데이터 추출하기

매크로기록기를 이용한 고급필터기능자동화

연봉일람표작성 마무리하기

1부. 연봉일람표자동화하기

2. 연봉일람표기능소개

연봉일람표는 지정한 조건에 맞는 직원 정보를 직원명부에서가져

오며, 연봉표에서 입력된공통 또는 개별 인상률을가져와새 연봉과

이전 연봉 그리고 인상률을 각각표시하게 된다.

인상률, 현재 연봉

직원 정보

1부. 연봉일람표자동화하기

요약

연봉일람표는 고급필터기능을이용하기 때문에직원명부에서필요

한 직원정보를 자유롭게가져와구성할 수있으며, 가져온정보를 바

탕으로근속년수와퇴직여부를표시한다.

연봉일람표는 연봉표에입력한연봉인상률에따라 새로계약을 체

결할 연봉 액수를 자동으로 표시한다.

1부. 연봉일람표자동화하기

3. 직원명부이해하기 (1)

직원명부는간이 인사기록부의성격을 갖고있는표로 부서, 직위, 
호봉, 주민등록번호, 입사일, 퇴사일, 연락처 등의 사원 개인정보를

담고 있다.

1부. 연봉일람표자동화하기

3. 직원명부이해하기 (2)

사원번호 계산

새로 입사한직원 정보를등록하면자동으로일련번호가부여된

다.

• =IF(LEN(B3)>0, COUNTA($B$3:B3), "")

주민등록번호 처리 (1)

주민등록번호뒷자리감추는 방법

• 주민등록번호는 "-" 구분자를 사용하지않고 입력한다. 

• 셀 서식의 "사용자 지정 숫자 서식"이용

• 000000- 0,, " ****** "

1부. 연봉일람표자동화하기

3. 직원명부이해하기 (3)

주민등록번호 처리 (2)

주민등록번호를 이용한 성별 계산방법

• 뒤 첫번째코드를 이용 (2로 나눠서 0이면 여자, 1이면 남자)

• =CHOOSE(MOD(MID(F3,7,1),2)+1, "여", "남")

주민등록번호를 이용한 나이 계산방법

• 뒤 첫번째코드와 앞의 두자리 코드를이용, 출생년도를 계산

• 나이 = 현재년도 – 출생년도 + 1

• =YEAR(TODAY())-
1*((INT(MOD(MID(F3,7,1)+1,10)/2)+18)&LEFT(F3,2))+1



엑셀 자동화 서식 만들기 (인사/총무편) 7

1부. 연봉일람표자동화하기

3. 직원명부이해하기 (4)

함수 (1)

IF

• 첫 번째 인수의결과값이 참이냐 거짓이냐에따라 참이면 두번째 인

수 부분이 실행되고, 거짓이면세 번째 인수부분이 실행되는함수

• =IF(조건, 참이면 실행, 거짓이면 실행)

LEN

• 셀에 입력된 문자열의 문자 개수를 세어주는함수

• =LEN(셀 또는텍스트)

CHOOSE

• 첫 번째 인덱스번호에 해당하는 값 을반환하여 주는 함수

• =CHOOSE(인덱스, 값1, 값2, 값3, …, 값29)

MOD

• 숫자를 나눌 값으로 나눈 다음남는 나머지 값을 돌려주는 함수

• =MOD(숫자, 나눌 값)

1부. 연봉일람표자동화하기

3. 직원명부이해하기 (5)

함수 (2)

MID, LEFT

• 텍스트를왼쪽에서지정한 문자수만큼 잘라내는함수는

LEFT, 중간부터 지정한 문자 수만큼잘라내는함수는 MID 이
다. 

• =MID(텍스트, 시작위치, 문자 수)

• =LEFT(텍스트, 문자수)

YEAR

• 날짜 일련번호를 받아 년도를 돌려주는 함수

• =YEAR(날짜일련번호)

TODAY

• 오늘 날짜를돌려주는함수

1부. 연봉일람표자동화하기

요약

직원의정보는 직원명부에서 관리된다.

계산할항목과 입력할 항목을 구분할수 있어야 표를 관리하기쉽다.

사원번호 : COUNTA 함수를 이용한 일련번호생성

주민등록번호 : 나이, 성별을 계산

1부. 연봉일람표자동화하기

4. 연봉표이해하기 (1)

연봉표는 직원들의현재연봉을기입해 놓고있는표로 부서,직위,이
름,연봉,인상률 등의정보를 담고있다.

공통인상분표는 직위별로 연봉을 일괄 인상하고자 할때 적용할 인

상률을기입해 놓는표이다.

1부. 연봉일람표자동화하기

4. 연봉표이해하기 (2)

직원명부에서 부서, 직위, 이름참조하기

직원명부의수정사항이연봉표에바로 반영되도록 하려면, 값을

복사해서사용하지말고, 참조해서사용해야한다.

부서 : =VLOOKUP($A4, 직원명부!$A$3:$D$22, 3,FALSE)

직위 : =VLOOKUP($A4, 직원명부!$A$3:$D$22, 4,FALSE)

이름 : =VLOOKUP($A4, 직원명부!$A$3:$D$22, 2,FALSE)

1부. 연봉일람표자동화하기

4. 연봉표이해하기 (3)

동적범위참조하도록이름 정의하기

연봉표

• 연봉표에새로운 직원이들어가도연봉일람표에서바로인식

할 수있도록범위를동적으로 참조

• =OFFSET(연봉표!$A$4,0,0,COUNTA(연봉

표!$A$4:$A$1000), 6)

이름

• 연봉일람표에서 연봉표의액수를가져오고자할 때, "이름"으
로 참조(MATCH) 하여 값을 가져오기때문에수식에서 사용하

기 위해 "이름"범위를 동적으로 참조

• =OFFSET(연봉표!$D$4,0,0,COUNTA(연봉

표!$D$4:$D$1000))

1부. 연봉일람표자동화하기

4. 연봉표이해하기 (4)

함수

VLOOKUP

• 찾을 값을 지정한 범위의 첫번째 열에서 찾아, 지정한 열번호에 존재

하는 값을 반환하는 함수

• =VLOOKUP(찾을 값, 범위, 열 번호, 옵션)

OFFSET

• 셀 또는 범위를이동하거나, 선택할 범위를 지정한 인수에 따라 열 또

는 행 방향으로확장하고자 할때 사용하는 함수

• =OFFSET(처음위치, 행 이동, 열 이동, 행 개수, 열 개수)

COUNTA

• 인수로 전달된범위 내의 빈셀을 제외한 모든셀의 개수를 세어 주는

함수

• =COUNTA(범위)

1부. 연봉일람표자동화하기

요약

연봉표이해하기

직원들의연봉을 기록하고, 인상률을기록할수 있는 표

직원명부의직원정보자료는 참조해서 사용해야데이터를두번

고치는불편함을 없앨 수있다.

동적범위를이름으로참조

수식을간결하고 이해하기 쉽게하는장점이있다.

데이터를추가해도, 수식변경 없이 바로 적용가능하다.

1부. 연봉일람표자동화하기

5. 연봉일람표작업순서

연봉일람표를 만드는 작업 순서를 다음과 같이 표시할 수있다.

1부. 연봉일람표자동화하기

6. 근속년수계산하기 (1)

근속년수는기업의경쟁력을 알수있는 하나의 척도로, 직원의입사

후 현재(오늘)까지 연속으로 근무한 기간을 "y년 m개월 d일" 형태로

표시한것이다.

= IF(AND(LEN(E12)>0,J12<>"퇴직"), 
DATEDIF(E12,TODAY(),"y")&"
년 "&DATEDIF(E12,TODAY(),"ym")&"개
월 "&DATEDIF(E12,TODAY(),"md")&"일", "")

1부. 연봉일람표자동화하기

6. 근속년수계산하기 (2)

함수

AND

• 전달된 인수의조건을 모두 만족하면 TRUE, 그렇지 않으면 FALSE 
를 반환하는 함수

DATEDIF

• 두 날짜 사이의차이를 계산하여 주는 함수

• =DATEDIF(시작일, 종료일, 계산옵션)

– 계산옵션

» y : 두 날짜사이의 년도 차를계산

» ym : 두 날짜가같은 년도에 있다고 가정, 월의차를 계산

» m : 두 날짜사이의 월의 차를계산

» md : 두날짜가같은 월에 있다고 가정, 일의차를 계산

» d : 두날짜 사이의 일의 차를계산

1부. 연봉일람표자동화하기

요약

DATEDIF 함수를 이용한근속년수구하기

입사일부터퇴직일또는오늘까지의근속기간을근속년수라고표

현한다.

이 항목으로기업의인적자원에대한경쟁력을가늠해 볼수있다.

구해진결과값은 "y년 m개월 d일" 형태로 표현된다.

1부. 연봉일람표자동화하기

7. 인상률에기반한연봉계산하기 (1)

연봉표에서직원의현재연봉과인상률을 가져와새로 지급할연봉

을 계산한다.

개별인상률은 직위별 공통 인상률에우선합니다.

인상률

• =IF(ISERROR(MATCH(D12,이
름,0)),"",IF(ISNUMBER(INDEX(연봉표,MATCH(D12,이
름,0),6)), INDEX(연봉표,MATCH(D12,이
름,0),6),VLOOKUP(C12,연봉표!$H$3:$I$16,2,FALSE)))

이전 연봉

• =IF(ISERROR(MATCH(D12, 이름, 0)), "", INDEX(연봉표, 
MATCH(D12, 이름, 0), 5))

신규 연봉 (십만 단위절사)

• =IF(LEN(H12)>0, ROUNDDOWN(H12*(1+I12), -5), "")

1부. 연봉일람표자동화하기

7. 인상률에기반한연봉계산하기 (2)

함수 (1)

ISERROR

• 에러가발생하면 TRUE, 에러가발생하지 않으면 FALSE 값을

반환한다.

INDEX

• 범위에서행과 열번호위치에 맞는 값을 반환하여주는 함수.  
주로 MATCH 함수와연결되어서외부에 존재하는데이터값

을 참조하고자 할때 사용된다.

• =INDEX(범위, 행 번호, 열번호)

ROUNDDOWN

• 값을 지정한단위에서절사하여주는함수

• =ROUNDDOWN(값, 단위)

• 단위는소수점 아래는 양수로, 소수점왼쪽은음수로 표시한다.

1부. 연봉일람표자동화하기

7. 인상률에기반한연봉계산하기 (3)

함수 (2)

VLOOKUP

• 찾을 값을 범위의 왼쪽 첫번째 열에서찾아, 범위의열 번호에

위치한값을돌려받는참조 함수

• =VLOOKUP(찾을 값, 범위, 열번호, 옵션)

– TRUE 또는생략 : 정확한값을 찾고, 없으면 찾을 값보다

작은 값중가장 큰값을찾아 준다.

– FALSE : 정확하게 일치하는 값만찾는다.  찾지 못하면

#N/A 오류값을 반환한다.

1부. 연봉일람표자동화하기

7. 인상률에기반한연봉계산하기 (4)

함수 (3)

MATCH

• 찾을 값이 지정한 범위의몇 번째행에존재하는지그 인덱스

값을 반환하여 주는함수

• =MATCH(찾을 값, 범위, 찾을 옵션)

– 찾을 옵션

»-1 : 일치하는값이 없다면근사값을찾는데, 조건은찾

을 값보다 작은 값중 최대값을 반환 (오름차순으로정

렬)

»0 :  정확하게일치하는첫번째 값을 찾는다.

»1 또는생략 : -1 옵션과동일하나찾을값보다큰 값중

최소값을반환 (내림차순으로 정렬)

1부. 연봉일람표자동화하기

요약

신규연봉계산하기

인상률을참조할 때, 개별인상률이존재하는지확인하고, 개별

인상률이존재하면직위별 공통인상률에 우선하여 처리한다.

현재 연봉을연봉표시트에서 참조한다.

신규 연봉을현재 연봉 * (1+인상률) 수식으로계산한다.

이 때, 십만단위는 절사한다. (ROUNDDOWN)

1부. 연봉일람표자동화하기

8. 퇴직현황표시하기

비고란에 퇴직 여부를 표시한다.

=IF(LEN(D12), IF(LEN(VLOOKUP(D12, 직원명

부!$B$3:$J$22, 9, FALSE)), "퇴직", ""), "")

1부. 연봉일람표자동화하기

요약

비고란에 퇴직 현황이 표시된다.

퇴직 현황은직원명부의퇴사일이기입되어있는지를 확인해존

재하면 "퇴직"이라는 값을표시한다.

퇴직현황이표시됨으로써, 퇴사한직원의 자료가 연봉일람표에나

타나는지여부를 확인해볼 수있다.

1부. 연봉일람표자동화하기

9. 고급필터를이용하여원하는데이터추출하기

조건표만들기

조건표의열이름과복사위치의열 이름은 원본표의 열이름과동

일해야한다.

같은 행의 조건은 AND 조건

다른 행의 조건은 OR 조건

1부. 연봉일람표자동화하기

요약

조건표는 고급필터의조건을 기입하기 위한표이다.

다음과같은조건을만족해야 한다.

조건표의열이름은원본테이블의열이름과일치해야 한다.

같은 행의 조건은 AND 조건

다른 행의 조건은 OR 조건

1부. 연봉일람표자동화하기

10. 매크로기록기를이용한고급필터기능자동화 (1)

자동화작업과정

조건표입력행 계산하기

• =LEN(B2)+LEN(C2)+LEN(D2) 

조건표동적범위 참조하기

• [삽입]-[이름]-[정의] 메뉴이용

• 이름 : "조건"

• 수식 : 
=OFFSET($B$1,0,0,COUNTIF($A$1:$A$4, ">=1")+1, 3)

매크로기록기를 이용해고급필터적용 과정기록하기

매크로기록기로 기록된코드 수정하기

• 조건표의범위를 위에서정의한이름으로 변경

"양식"도구모음의 "버튼"컨트롤에매크로 연결하기

1부. 연봉일람표자동화하기

10. 매크로기록기를이용한고급필터기능자동화 (2)

함수

COUNTIF

• 지정한범위내에서조건에 맞는셀의개수를세어반환해주

는 함수

• =COUNTIF(범위, 조건)

1부. 연봉일람표자동화하기

요약

매크로기록기를 이용한고급필터기능 자동화

조건표를동적 범위로 참조 (이름사용)

고급필터를이용한데이터 추출과정을 매크로기록기로기록

기록된매크로 코드중에서 조건표를지정한범위를 찾아, 정의해

놓은 이름으로 수정

"양식"도구모음의 "버튼"컨트롤을삽입해 기록된매크로연결

1부. 연봉일람표자동화하기

11. 연봉일람표작성마무리하기

조건부서식을 이용해 연봉, 인상률 정보를 감추는방법

연봉 및인상률 정보는 사내 비밀정보에 해당하므로 필요한 경우

에만 보여질수 있도록 조건부 서식을이용하여값을 감추는 방법

을 사용한다.

인쇄영역설정하기

조건표등이나타나지않고, 필요한 "연봉일람표"부분만출력되

도록 [파일]-[인쇄 영역]-[인쇄 영역 설정] 메뉴를클릭해 인쇄 영

역을 설정 한다.

1부. 연봉일람표자동화하기

요약

조건부서식을 이용하면필요한부분의 값만감출수 있다.

하나의시트에 여러개의표가 존재할때, [파일]-[인쇄 영역]-[인쇄

영역 설정] 메뉴를 이용하면 필요한 부분만 인쇄할수 있다.

1부. 연봉일람표자동화하기

12. 1부요약

연봉일람표자동화하기

고급필터를이용해조건에 맞는직원들의 데이터를 추출할 수있

다.

근속년수를자동으로계산해 준다.

개별 또는 공통 인상률에따른 새연봉을 자동으로계산해준다.

퇴직 여부를연봉일람표에서 바로확인할 수있다.

매크로기록기를 이용해 데이터를추출하는 과정을기록하고편집

해 업무를 자동화한다.

2부. 연봉계약서자동화하기

최 준 선

오피스튜터강사

2부. 연봉계약서자동화하기

1. 도입말 (1)

강좌목적

연봉일람표를참조해직원들의 개별 연봉계약서를한번에출력할

수 있다.

강좌소개

워드의 "편지병합"기능을이용해서하나의 기본 서식으로여러

장의 문서를출력하는방법을 알아본다.

"병합 필드"의 서식지정방법과계산식을 사용하는 방법을 알아

본다.

2부. 연봉계약서자동화하기

1. 도입말 (2)

강좌내용

연봉계약서기능 소개

연봉일람표이해하기

연봉계약서작업 순서

워드의 "편지병합" 기능을이용, 엑셀 파일과 연결하기

"병합 필드"의 "숫자서식" 지정하기와계산식작성하기

2부. 연봉계약서자동화하기

2. 연봉계약서기능소개

서식기능확인하기

연봉계약서구성

• 계약기간및연봉

• 지급 방법

• 계약 갱신

• 비밀 유지

엑셀의데이터가 워드의문서와

정확하게연결된다.

2부. 연봉계약서자동화하기

요약

편지병합 기능을 이용해엑셀의데이터를 하나의서식 파일에연결

해 모두인쇄할 수있다.

병합필드를선택하면짙은 회색테두리로 표시된다.

연결된엑셀파일에존재하지 않는 항목이라도, 기존 항목으로계산

이 가능한 것(예 : 연봉 인상분)이면 계산필드를이용해 표시할 수있

다.

2부. 연봉계약서자동화하기

3. 연봉일람표이해하기

연봉일람표에 구성된 내역을 워드의 "편지 병합" 기능을 이용해연

봉계약서에반영하여출력하기위해서는 고급 필터로 추출된일람표

전체를이름으로 참조해야 한다.

이름 : 연봉일람
범위 : 연봉일람표 출력 부분

2부. 연봉계약서자동화하기

요약

엑셀파일을원본 데이터로 "편지병합"을 하는경우, 시트전체또

는 정의된 이름을 사용해연결할수 있다.

이 때, 정의된이름이동적으로 범위를참조하도록수식이구성된 경

우에는워드의 "편지병합"기능을 설정할 때정의된이름이연결할

테이블에나타나지않는다.

범위를이름으로 지정할때는 반드시제목 행을 포함해야된다.

2부. 연봉계약서자동화하기

4. 연봉계약서작업순서

작업순서

2부. 연봉계약서자동화하기

5. 워드의 "편지병합" 기능을이용, 엑셀파일과연결하기

작업과정

[도구]-[편지및 우편물]-[편지 병합] 메뉴를 클릭

• 마법사 1단계 : 문서종류선택 -> "편지" 선택

• 마법사 2단계 : 시작문서선택 -> "현재문서" 선택

• 마법사 3단계 : 엑셀파일선택 / 연결

• 마법사 4단계 : 편지작성 -> 병합된필드를 문서와 연결

• 마법사 5단계 : 편지미리보기

• 마법사 6단계 : 병합완료

2부. 연봉계약서자동화하기

요약

"편지 병합" 기능을이용해 엑셀데이터를 워드 문서에 연결

엑셀 데이터의 경우, 이름또는 시트명을 선택해 데이터를연결할

수 있다.

데이터를가져올 때, 조건을 지정해 조건에 맞는 데이터만가져올

수 있다.

"편지 병합" 마법사에서필드를연결하지 않고, 마법사를종료한

다음 "편지병합" 도구모음의 "병합필드삽입" 버튼을눌러원하

는 위치에 항목을 연결할수 있다.

2부. 연봉계약서자동화하기

6. "병합 필드"의 "숫자서식" 지정하기와 계산식 작성하기 (1)

연봉액수에천 단위구분기호 표시하기

워드와병합된 필드는 모두 "텍스트"로 인식되기때문에, 엑셀에

서 지정한 숫자 서식이 표시되지않는다.

작업 과정

• "연봉" 표시영역 선택, "단축메뉴" 호출 후, "필드코드 토글" 
메뉴 클릭

• 다음과같이 "병합 필드" 값이 바뀌게 되면 "필드명" 뒤에 "₩" 
삽입 후숫자서식을입력한 다음 F9 키를 눌러주면된다.

{ MERGEFIELD "연봉" }

{ MERGEFIELD "연봉" ₩ #,### }

2부. 연봉계약서자동화하기

6. "병합 필드"의 "숫자서식" 지정하기와 계산식 작성하기 (2)

인상률을백분율로표시하고, 연봉 인상분 계산하기

"병합 필드"에서는백분율 서식을 지정할 수없으므로, 계산식을

이용해백분율을 계산해야 한다.

작업 과정

• "병합 필드" 영역 선택, "단축메뉴" 호출 후, "필드편집" 메뉴

클릭

• "필드" 대화상자의 "필드이름" 항목에서 "=(Formula)" 선택

• "병합 필드" 영역에 { = }가 삽입된다.

• "=" 다음에 "편지 병합" 도구모음의 "병합 필드삽입" 메뉴를

클릭하여원하는 필드를집어넣고사칙연산자(+,-,*,/)를 이용

해 계산을 수행하면된다.

{ = 0.4455444 * 100 } % : 인상률

{ = 연봉 – 이전연봉₩ #,### } : 인상분

2부. 연봉계약서자동화하기

요약

엑셀의항목을 워드와 연결하는경우, 텍스트로인식되기때문에 엑

셀에서지정한 서식은 모두 무시된다.

워드의 "병합필드"에서는별도의숫자서식을지정할 수있다.

지정할수있는 숫자서식중 백분율은지원하지않는다.

원본테이블에 존재하지않는 항목도계산할수 있으면 "병합필드"
에 수식을 이용해 계산해넣을 수있다.

2부. 연봉계약서자동화하기

7. 2부요약

연봉계약서자동화하기

반복적인서식을 출력하는 작업이 필요하다면워드의 "편지병

합" 기능을이용하는것이효율적이다.

"병합 필드"에 대해이해해 두어야 하는 것들

• 엑셀에서지정한 숫자 서식을 가져다사용할수가없다.

• 적절한숫자서식을지정할 수있다.

• 숫자 서식에백분율은지원하지않기때문에계산식을 이용해

백분율로표시할 수있다.

• 원본 테이블의 항목간계산 결과를 "병합 필드"에표시할 수

있다.

3부. 인건비분석표자동화하기

최 준 선

오피스튜터강사

3부. 인건비분석표자동화하기

1. 도입말 (1)

강좌목적

연봉일람표를이용해쉽고빠르게회사의 인건비를분석할수 있

다.

강좌소개

인건비분석표를 작성하는데 필요한 참조 데이터는무엇이고, 인
건비를분석하기 위해 어떠한 항목을계산해야하는지 알아본다.

3부. 인건비분석표자동화하기

1. 도입말 (2)

강좌내용

인건비분석표기능소개

연봉일람표이해하기

• 7월 퇴직자 및현행근무자리스트추출하기

인건비분석표작업순서

분석 월입력하기

전월 분석 항목 계산하기

금월 분석 항목 계산하기

조건부서식을 이용해 조건에 맞는 데이터 표시하기

매출증감률및 인건비증감률 계산하기

3부. 인건비분석표자동화하기

2. 인건비분석표기능소개

인건비분석표는 기업의인건비가부서별로어떻게 들어가고있는지, 
회사 평균 연봉에 비해 어느 부서가 더높은 연봉을받는지, 그리고

회사의매출대비 인건비비율은어떠한지 등을 알려주는표이다.

3부. 인건비분석표자동화하기

요약

인건비분석표는 매월 인건비 변동 내역 그리고 매출 대비인건비 비

율을 점검함으로써효율적인 인원관리 계획을 수립하도록도와준다.

매월새로운분석작업을하려면다음과 같은방법을 사용한다. 

연봉일람표를그 달퇴사자가 함께표시되도록구성한다.

인건비분석표의 월별날짜를 수정한다.

매출 내역을수정한다.

3부. 인건비분석표자동화하기

3. 연봉일람표이해하기 (1)

7월(금월) 퇴직자와현재근무자의연봉일람표구성하기

조건표에삽입할 조건

• 퇴사일

="="  또는 ‘=

=">=" & DATE(2005, 7, 1)

3부. 인건비분석표자동화하기

3. 연봉일람표이해하기 (2)

이름정의하기

월별 인원 및연봉 총액을계산하기위해 연봉일람표의 전체 범위

를 선택하여열 이름을 사용해 이름을지정한다.

• B6:J111 범위를 선택한다.

• [삽입]-[이름]-[만들기] 메뉴를 클릭

• "첫 행" 항목만 선택하고 <확인> 버튼 눌러 종료

3부. 인건비분석표자동화하기

요약

인건비분석표는 연봉일람표를참조해 구성하는데, 연봉일람표는 현

재 재직중인직원과분석할 월의퇴사자가 포함되도록 구성한다.

인건비분석표의 수식작성을 쉽고 간결하게하기위해 복사해온 연

봉일람표데이터 범위를이름으로정의한다.

3부. 인건비분석표자동화하기

4. 인건비분석표작업순서

작업순서

3부. 인건비분석표자동화하기

5. 분석월입력하기

분석월은직원과 연봉을계산하기위해 정확하게입력해야한다.

분석 월의 1일에 해당하는값을 입력

셀 서식(CTRL+1) 대화상자 호출

• "표시형식" 탭에서 "사용자 지정" 선택

– yyyy"년도" mm"월"

• "맞춤" 탭에서 "방향"을 "세로"로 지정

3부. 인건비분석표자동화하기

요약

분석월을입력할 때, 분석할 월의 1일을입력해직원과 연봉계산할

때 참조한다.

셀 서식을지정해 입력된값을 월로 표시한다.

3부. 인건비분석표자동화하기

6. 전월분석항목계산하기 (1)

직원

전월 직원 = 부서인원 – 7월 입사자

=COUNTIF(부서, C14)-SUMPRODUCT((부서=C14)*(입사

일>=$O$8))

연봉

전월 연봉 = (부서전체연봉 – 7월 입사자연봉)/12

=(SUMIF(부서, C14, 연봉)-SUMPRODUCT((부서=C14)*(입사

일>=$O$8), 연봉))/12

1인당연봉

1인당 연봉 = 연봉 / 인원

=H14/F14

3부. 인건비분석표자동화하기

6. 전월분석항목계산하기 (2)

함수

COUNTIF

• 조건에맞는셀의 개수를세어 주는 함수

• =COUNTIF(범위, 조건)

SUMIF

• 조건에맞는범위에대응하는 범위의합계를낸다.

• =SUMIF(조건범위, 조건, 합계범위)

SUMPRODUCT

• 인수로전달된 배열을 각각 곱하고, 곱해진 값을 모두 더해주

는 함수

• =SUMPRODUCT(배열1, 배열2, …, 배열30)

• 여기서는조건이 하나 이상인 경우의합을 계산

• =SUMPRODUCT(조건1*조건2*…*조건N, 합계범위)

3부. 인건비분석표자동화하기

요약

전월인건비를 분석하기위해 다음과같은 수식을작성한다.

인원 = 전체인원 – 금월(7월)입사자

연봉 = (전체연봉 – 금월(7월)연봉)/12

1인당연봉 = 연봉/인원

3부. 인건비분석표자동화하기

7. 금월분석항목계산하기

직원

금월 직원 = 부서인원 – 퇴사자

=COUNTIF(부서, P14)-SUMPRODUCT((부서=P14)*(비고="퇴
직"))

연봉

금월 연봉 = (부서전체연봉 – 퇴직자연봉)/12

=(SUMIF(부서, P14, 연봉)-SUMPRODUCT((부서=P14)*(비고

="퇴직"), 연봉))/12

1인당연봉

1인당 연봉 = 연봉 / 인원

=U14/S14

3부. 인건비분석표자동화하기

요약

금월인건비를 분석하려면 다음과 같은 수식작성이 가능하다.

인원 = 전체인원 – 금월(7월)퇴사자

연봉 = (전체연봉 – 금월(7월)퇴사자연봉)/12

1인당연봉 = 연봉/인원

3부. 인건비분석표자동화하기

8. 조건부서식을이용해조건에맞는데이터표시하기

조건부서식지정하기

직원 (최대직원 보유부서표시)

• =$G14=MAX($G$14:$G$31)

연봉 (전체연봉대비가장많은 비율 차지하는부서표시)

• =$K14=MAX($K$14:$K$31)

1인당 연봉 (100% 를초과하는 곳과 가장 작은연봉을 받은부서

표시)

• =$N14>1

• =$N14=MIN($N$14:$N$31)

3부. 인건비분석표자동화하기

요약

표의가독성을 높이기 위해 조건부서식을 이용해관심항목에별도

의 서식을 지정한다.

가장 많은 직원을 보유한부서

가장 많은 연봉을 지급한부서

1인당 연봉이사내 평균 연봉을 초과하는 부서

1인당 연봉이가장 적은 부서

3부. 인건비분석표자동화하기

9. 매출증감률및인건비증감률계산하기

매출입력, 매출대비인건비 비율계산하기

매출대비인건비 비율 : 연봉/매출

매출증감

매출증감 = (금월매출 – 전월매출)/전월매출

인건비증감 = (금월인건비 – 전월인건비)/전월인건비

3부. 인건비분석표자동화하기

요약

매출증감과인건비증감을동시에표기함으로써, 인건비의비용지출

부분이너무과도한지아닌지를표기한다.

3부. 인건비분석표자동화하기

10. 3부요약

인건비분석표 자동화하기

기업은이익을 극대화하기 위해매출을 높이고 비용은 낮추는방

법을 선호하는데, 이서식은 매출대비인건비의비율을 계산해줌

으로써매출과 비용의 적절한 균형을유지하도록도와준다.

매월 데이터를 UPDATE 하기 위해서는 다음과 같은 작업이 필요

하다.

• 연봉일람표에서 당월퇴직자를 포함한전체 직원 리스트를뽑

는다.

• 날짜를수정한다.

• 매출액을기재한다. 

4부. 팀인사고과표자동화하기

최 준 선

오피스튜터강사

4부. 팀인사고과표자동화하기

1. 도입말 (1)

강좌목적

인사고과작업을 위해 관리자가작업해야 할부분을 최소화하고, 
평가한부분의 통계를 바로 보여줌으로써 관리자의 평가작업을

지원한다.

강좌소개

고급 필터를이용하여중복되지않은값을 추출하는 방법에 대해

알아본다.

조건을만족하는 리스트를 다른표에서 참조하려고 할때, 어떤수

식을 사용하는지 알아본다.

평가항목별로값을선택하면, 항목/인원별로 평가한값이 바로 집

계되는서식을 작성한다.

4부. 팀인사고과표자동화하기

1. 도입말 (2)

강좌내용

팀 인사고과표 기능소개

직원명부이해하기

팀 인사고과표 작업순서

중복된데이터에서고유항목 추출하기(고급 필터)

병합된셀에서 배열수식작성하기

수식을간소화하기위한조건부서식사용법

유효성검사를 이용해 채점 항목표시하기

항목/인원별평균 계산하기

상대평가를위한 순위 계산하기

4부. 팀인사고과표자동화하기

2. 팀인사고과표기능소개 (1)

인사고과표는, 회사가 정한 객관적인지표로구성원들을평가하여

그 결과를 인사고과에반영하기위한자료이다.

제공된인사고과표는누구나 쉽고 편리하게작성할 수있도록자동

화된 기능을제공한다.

팀을 선택하면 팀구성원(팀원)이 자동적으로열 목록에 표시된다.

평가 항목에해당하는점수를 목록에서 1~5까지의점수로채점한

다.

자동으로항목/인원별평균이 구해지며, 하단에 순위가 표시된다.

4부. 팀인사고과표자동화하기

2. 팀인사고과표기능소개 (2)

4부. 팀인사고과표자동화하기

요약

팀 인사고과표는 관리자(팀장)가 구성원(팀원)에 대한 평가를 쉽고

빠르게수행할 수있도록지원하는자동화된서식이다.

팀 선택기능

팀원 자동 표시 기능

평가 값선택기능

자동 집계 기능

순위 표시 기능

4부. 팀인사고과표자동화하기

3. 직원명부이해하기

직원명부 중에서 팀, 직책, 이름, 입사일항목만을가져온 축소판직

원명부이다.

인사고과표에서 팀원 이름을 참조하는 수식을 작성할 때, 수식을 간

결하고이해하기 쉽게 하기 위해직원명부의각 열을 제목행으로 이

름 정의해 놓는다.

4부. 팀인사고과표자동화하기

요약

인사고과표에서 참조하는 직원명부의 각열을수식 작성시간결하

고 이해하기쉽게 이름으로 정의한다.

직원은항상입사 또는 퇴직의 경우가발생하기때문에, 직원명부 시

트를 원본 파일에 연결하여 사용하는경우라면이름을 정의할때, 동
적 범위로 참조하는것이바람직하다.

4부. 팀인사고과표자동화하기

4. 팀인사고과표작업순서

자동화서식을 구성하기위한 방법을다음 순서대로 진행한다.

4부. 팀인사고과표자동화하기

5. 중복된데이터에서고유항목추출하기(고급필터)

팀 인사고과표에서팀 명이 중복되지않게 나타나게 하려면?

이름 정의 : 팀명

4부. 팀인사고과표자동화하기

요약

인사고과표에서 팀명을선택할때 중복된 값이 표시되지않게하기

위해, 직원명부의 팀이름범위에서고유 항목만 [고급 필터]를이용

해 추출해 낸다.

추출된범위를 "팀명"이란 이름으로정의해둔다.

인사고과표의팀 선택 목록을 지정할 때, 유효성검사에서 해당 이

름을 사용하기 위함이다.

4부. 팀인사고과표자동화하기

6. 병합된셀에서배열수식작성하기 (1)

팀을목록에서 선택

유효성검사이용해정의해 둔이름("팀명")을 목록으로 지정

팀원을자동표시

병합된셀에서는 배열 수식을 사용할수 없다.

셀에 배열 수식을 입력한다음 병합하면, 병합된 셀에 배열수식을

사용할수가있다.

=INDEX(이름, SMALL(IF((팀=$B$2)*(직책<>"팀장"), 
ROW(INDIRECT("1:" & COUNTA(이름)))), 
CEILING(COLUMN(A1), 2)/2))

배열 수식이므로 수식 작성 후 CTRL+SHIFT+ENTER 키로 수식

을 입력한다.

4부. 팀인사고과표자동화하기

6. 병합된셀에서배열수식작성하기 (2)

함수

SMALL

• 지정한범위또는 배열에서 N 번째로작은 값을 돌려준다.

• =SMALL(범위, N)

ROW, COLUMN

• 인수로전달된 셀의행번호(ROW), 열번호(COLUMN)를 반

환한다.

INDIRECT

• 텍스트문자열로 지정한참조를인식한다.

CEILING

• 첫 번째인수의 숫자를 두번째 인수의배수로올림처리한수

를 반환한다.

• =CEILING(숫자, 숫자)

4부. 팀인사고과표자동화하기

요약

팀 이름을목록에서선택할 수있도록유효성검사기능을이용한다.

서식을만들다 보면병합된 셀을처리하는 경우가발생하는데, 이런

경우 배열 수식을 입력할수 없어불편하다.

이 때, 병합된셀을해제하고 배열수식을 작성한 다음 다시병합을

하면 배열 수식을 병합된셀에서사용할 수가있다.

4부. 팀인사고과표자동화하기

7. 수식을간소화하기위한조건부서식사용법

수식이복잡해지는이유중의 하나가수식에에러처리 루틴을삽입

하기 때문이다.  에러 처리루틴이란 다음과 같은 것을 의미한다.

=IF(ISERROR(수식), 에러처리수식, 수식)

조건부서식을 이용해 에러가 발생할셀을 미리 지정함으로써에러

발생 시배경색과 글꼴색을 동일하게하여 에러가화면에표시되지

않도록할 수있다.

4부. 팀인사고과표자동화하기

요약

셀에입력되는 수식에 에러가 발생하는 경우, 에러처리 루틴을삽입

해 수식을 복잡하게하지않고 조건부서식을이용해 수식을 간소화

할 수있다.

수식이입력될 전체범위를 선택한다.

[서식]-[조건부 서식] 메뉴를 실행한다.

수식은 =ISERROR(첫 번째 선택셀)

서식을지정할 때글꼴색을 배경색과일치시킨다.

4부. 팀인사고과표자동화하기

8. 유효성검사를이용해채점항목표시하기

채점할항목을 미리목록으로 지정한다음 선택하는 동작만으로 평

가가 가능하도록 할수있다.

평가 값을 입력할 범위(G12:Z33)를 선택한다.

[데이터]-[유효성 검사] 메뉴를 클릭한다음, "데이터유효성" 대
화 상자에서다음과같이설정한다.

4부. 팀인사고과표자동화하기

요약

유효성검사의 목록기능을 이용하면, 셀에 입력될값을 목록의값으

로 제한하고입력할값을선택 가능하게 하는효과가 있다.

4부. 팀인사고과표자동화하기

9. 항목/인원별평균계산하기

평가항목/인원별 평가 작업을 할때, 평가한 값의 평균이 항목별인

원별로표시되어 지도록 AVERAGE 함수를이용해 계산한다.

=IF(ISERROR(AVERAGE(G12:G33)), "", 
AVERAGE(G12:G33))

종합평가계산하기

평균을계산할 때에러처리 루틴을 사용하고조건부 서식을 이용

하지 않은 것은, 바로 "종합 평가" 항목이 존재하기때문이다.

• =IF(ISERROR(AVERAGE(G10:Z11)), "", 
AVERAGE(G12:G33))

4부. 팀인사고과표자동화하기

요약

평가항목별로 관리자가점수를매기면 바로집계가 되어나타나도

록 항목/인원별 평균을 표시한다.

4부. 팀인사고과표자동화하기

10. 상대평가를위한순위계산하기

평가는관리자에 따라 점수가 항상 달라질 수있으므로, 관리자가 평

가한 값자체보다 상대적인 평가가 더중요하게된다.

그러므로 RANK 함수를이용하여순위를 구한다음, 인사고과에순

위를 반영해평가를한다.

이러한방식은 관리자의변동이생겨도 항상일정한 평가가 가능하

다는 장점이있다.

=RANK(G10, $G$10:$Z$10) 

RANK 함수는 순위를 구하는 함수로구문은다음과 같이사용된

다.

• RANK(값, 범위, 정렬) 

– 정렬

– 0 또는생략 : 내림차순

– 다른 값 : 올림차순

4부. 팀인사고과표자동화하기

요약

인사고과의경우 점수를매기는사람의 성향에 따라 점수가 달라지

므로 정확한인사고과를위해서는매겨진 점수의순위를매겨상대평

가로 구성원을 평가하는것이 바람직하다.

4부. 팀인사고과표자동화하기

11. 4부요약

팀 인사고과표 자동화하기

모든 작업이사용자의입력 없이마우스 클릭만으로 처리된다.

팀을 선택하면 팀원이 자동적으로표시된다.

평가 항목별로 점수를 선택하여채점하면 평가 항목별, 인원별점

수 평균이 바로 나타나고, 팀의 종합 평가 점수가 표시된다.

인원별평균점수로순위를 매겨, 인사고과가상대평가로진행될

수 있도록 지원한다.

5부. 경력증명서자동화하기

최 준 선

오피스튜터강사

5부. 경력증명서자동화하기

1. 도입말 (1)

강좌목적

경력증명서작성을위한참조표에대해 이해하고, 간단한확인작

업을 거쳐 바로 경력증명서를 뽑을 수있는 자동화된 서식을 작성

한다.

강좌소개

주민등록번호를 참조하고직원명부와부서이동기록표를 이용해

경력증명서를자동으로 구성하여본다.

일률적이지않은 병합 셀을 대상으로배열 수식을사용하기위한

방법을알아본다.

5부. 경력증명서자동화하기

1. 도입말 (2)

강좌내용

경력증명서기능 소개

직원명부이해하기

부서이동기록표 이해하기

경력증명서작업 순서

주민등록번호목록만들기

병합된셀에배열 수식을이용하는또다른 방법

여러 경력사항에서근속년수 계산하기

지원되지않는 원문자사용하는방법

5부. 경력증명서자동화하기

2. 경력증명서기능소개

경력증명서에는, 입력된주민등록번호(KEY)에 따라사원정보및

부서이동경력이 자동으로 표시된다.

5부. 경력증명서자동화하기

요약

경력증명서는 퇴직자의주민등록번호를 이용해 직원명부와부서이

동기록표에서각각인적사항과 경력 사항을 가져와자동으로표시해

준다.

5부. 경력증명서자동화하기

3. 직원명부이해하기

경력증명서에서 사용할직원명부의부분 또는 전체를 이름으로 정

의한다.

정의할이름

• 주민등록번호 : 경력증명서에서 직원 여부를 판단하고 KEY로
사용할목적으로 이름으로 정의, 동적범위 참조

• 직원명부 : 직원인 경우, 이름, 주소, 사원번호를참조하기 위

해 전체범위를 이름으로정의, 동적 범위 참조

5부. 경력증명서자동화하기

요약

경력증명서에서 직원명부의 주민등록번호를목록에서선택할 수있

도록 주민등록번호범위를 이름으로정의한다.

주민등록번호가 존재할경우, 사원번호, 이름, 주소정보를참조하기

위해 직원명부를 이름으로 정의한다.  

이 때, 정의할이름은데이터의 추가/삭제가 자유롭도록 동적 범위

참조방식으로정의한다.

5부. 경력증명서자동화하기

4. 부서이동기록표이해하기

첫 인사 조치로 부서를 배치 받은이후부서 이동의결과로부서가

바뀐 경우 해당 내용을 기록하는표로, 경력증명서의경력사항을표

시하는데도움이 된다.

5부. 경력증명서자동화하기

요약

부서이동기록표는부서이동 경력을표시함으로써해당사원의경력

을 내부적으로 관리할 수있고, 향후 경력증명서에이를 표기해사원

의 경력을 객관적으로증명할 수있다.

5부. 경력증명서자동화하기

5. 경력증명서작업순서

자동화서식을 구성하기위한 방법을다음 순서대로 진행한다.

5부. 경력증명서자동화하기

6. 주민등록번호목록만들기

직원명부에서이름으로 정의한 주민등록번호를, 경력증명서에서 유효성검

사의 대상 목록으로 사용할 수있도록 한다.

사원번호 참조 : =IF(LEN($N$7)>0, INDEX(직원명부, MATCH(N7, 주민등

록번호, 0), 1), "")

이름 참조 : =IF(LEN(사번)>0, VLOOKUP(사번, 직원명부, 2, FALSE), "")

주소 참조 : =IF(LEN(사번)>0, VLOOKUP(사번, 직원명부, 11, 
FALSE), "")

5부. 경력증명서자동화하기

요약

유효성검사의 목록을 지정하면다음과 같은장점이 있다.

값을 입력하지 않고, 입력할 값을선택할 수있다.

목록에존재하지 않는 값은 입력되지않는다.

주민등록번호로 사원 번호를 참조해와서 직원명부의 자료를참조

하도록한다.

5부. 경력증명서자동화하기

7. 병합된셀에배열수식을이용하는또다른방법 (1)

병합된셀이인사고과표와는 달리일정하지 않다면, 별도의참조표

를 만든다음, 해당 참조표를 재참조하는 방식을 사용하면된다.

5부. 경력증명서자동화하기

7. 병합된셀에배열수식을이용하는또다른방법 (2)

수식

다른 시트에서 조건에 맞는 데이터를여러 개가져오기 위해서는

배열 수식을사용해야한다.

아래 수식 작성 후, CTRL+SHIFT+ENTER키로 입력해야 한다. 

=INDEX(부서이동기록표, SMALL(IF(사원번호=사번, 
ROW(INDIRECT("1:" & COUNT(사원번호)))), ROW(A1)), 
V$20)

5부. 경력증명서자동화하기

요약

일정하게 병합되지않은병합 셀에 배열 수식을 사용하려면?

별도의참조표를 만들어배열 수식을사용한다.

참조표의값을 병합셀에서 다시참조한다. 

5부. 경력증명서자동화하기

8. 여러경력사항에서근속년수계산하기

경력사항이여럿 존재할경우, 근속년수를 계산하려면 입사일과 퇴

사일을대상으로 DATEDIF 함수를사용하면 된다.

=DATEDIF(MIN(C13:F19), MAX(G13:J19), "y") & "년 " & 
DATEDIF(MIN(C13:F19), MAX(G13:J19), "ym") & "개월 " & 
DATEDIF(MIN(C13:F19), MAX(G13:J19), "md") & "일"

함수

MAX

• 인수로전달된 값에서 최대값을반환하여 주는 함수

MIN

• 인수로전달된 값에서 최소값을반환하여 주는 함수

5부. 경력증명서자동화하기

요약

경력사항이여럿 존재하면 MAX, MIN 함수를이용하여 입사일과 퇴

사일을계산하여 DATEDIF 함수를사용해 근속년수를 계산한다.



엑셀 자동화 서식 만들기 (인사/총무편) 58

5부. 경력증명서자동화하기

9. 지원되지않는원문자사용하는방법

지원되지 않는 원문자를사용하려면, 워드의 "원문자" 기능을이용

한다.

워드를실행하고, [서식]-[문자 모양]-[원 문자] 메뉴를 클릭해 사

용하고자하는 원문자를생성한다.

생성된원 문자를 복사 한다음, 엑셀 파일을 열고 사용하기원하

는 셀을선택한 다음마우스 오른버튼을 눌러 '단축메뉴'에서 "선
택하여붙여넣기"를 클릭한다.

"선택하여붙여넣기" 대화상자에서 "그림(확장 메타파일)"을 선

택한 다음 <확인> 버튼을 누르면 된다.

5부. 경력증명서자동화하기

요약

지원되지 않는 원문자를엑셀에서사용하려면, 워드의 "원문자" 기
능으로원 문자를 생성한후 [복사]-[선택하여붙여넣기] 를이용해 엑

셀 시트에 이미지로가져올 수있다.

5부. 경력증명서자동화하기

10. 5부요약

경력증명서자동화하기

직원명부와부서이동기록표를 참조해서 경력증명서 서식구성

[유효성 검사] 메뉴를이용해 주민등록번호 목록 만들기

주민등록번호를 기준으로사원번호참조

사원번호로이름, 주소 참조하기

셀의 개수가일정하지않은 병합셀에서 배열수식을 사용하기위

해 참조표를이용하는방법 소개

여러 경력사항에서근속년수 계산하기

지원되지않는 원문자사용하는방법


